

Bounds on Magnitudes of Range Polynomial Coefficients under Max-Throttle Constraint

We would like to bound the search space for polynomial range (i.e. phase) models.

Bounds on the Taylor series approximation error become unwieldy at high polynomial orders: the high-order derivatives of the throttle function are not nicely bounded.

Another way of bounding the coefficients of the range polynomial is to impose a bound on the magnitude of the throttle function. Suppose that the throttle function plays the role of \ddot{x} , so that its double integral is the (not necessarily polynomial) range function. Let us consider the linear minimum mean squared error range polynomial corresponding to this range function.

To obtain the LMMSE estimate, we write

$$\min_{c_n} \int_0^T dt \left| x[t] - \sum_{n=0}^d c_n t^n \right|^2, \text{ with normal equations}$$

$$0 = \int_0^T dt \left(x[t] - \sum_{m=0}^d c_m t^m \right) \begin{pmatrix} 1 & t & \dots & t^d \end{pmatrix}^T, \text{ or}$$

$$\int_0^T dt x[t] \begin{pmatrix} 1 & t & \dots & t^d \end{pmatrix}^T = \sum_{m=0}^d c_m \begin{pmatrix} \frac{T^{m+1}}{m+1} & \frac{T^{m+2}}{m+2} & \dots & \frac{T^{m+d+1}}{m+d+1} \end{pmatrix}^T$$

Let us non-dimensionalize the problem by taking $\{t \rightarrow u T, T^{m-1} c_m \rightarrow b_m\}$.

$$\int_0^1 du x[u T] \begin{pmatrix} 1 & u & \dots & u^d \end{pmatrix}^T = \sum_{m=0}^d b_m \begin{pmatrix} \frac{1}{m+1} & \frac{1}{m+2} & \dots & \frac{1}{m+d+1} \end{pmatrix}^T$$

$$= \begin{pmatrix} \frac{1}{1} & \frac{1}{2} & \dots & \frac{1}{d+1} \\ \frac{1}{2} & \frac{1}{3} & & \\ \vdots & & \ddots & \vdots \\ \frac{1}{d+1} & \dots & \frac{1}{2d+1} \end{pmatrix} b$$

$$= \text{HilbertMatrix}[d+1] \cdot b$$

Now we may solve for the vector of non-dimensionalized coefficients b :

$$b = \int_0^1 x[u T] P_d[u] du, \text{ where we introduce}$$

$$P_d[u] = \text{Inverse}[\text{HilbertMatrix}[d+1]] \begin{pmatrix} 1 & u & \dots & u^d \end{pmatrix}^T, \text{ and it will be helpful to define}$$

$$Q_d[u] = \int_0^u P_d[u'] du'$$

$$R_d[u] = \int_0^u Q_d[u'] du'$$

So far, we have obtained b in terms of $x[t]$. Because our throttle constraints are most naturally expressed instead in terms of $\ddot{x}[t]$, it would be helpful to re-express b in terms of $\ddot{x}[t]$. We can accomplish this by integrating by parts two times.

$$\begin{aligned}
 b &= x[T] Q_d[1] - T \int_0^1 Q_d[u] \dot{x}[u T] du & (U = x) \\
 &= x[T] Q_d[1] - T \dot{x}[T] R_d[1] + T^2 \int_0^1 R_d[u] \dot{x}[u T] du & (U = \dot{x}) \\
 &= x_{\text{final}} Q_d[1] - T v_{\text{final}} R_d[1] + T^2 \int_0^1 R_d[u] \dot{x}[u T] du \\
 |b| &\leq |x_{\text{final}}| \times |Q_d[1]| + T |v_{\text{final}}| \times |R_d[1]| + T^2 |\dot{x}_{\text{max}}| \times \int_0^1 |R_d[u]| du \\
 &= A_d |x_{\text{final}}| + B_d T |v_{\text{final}}| + C_d T^2 |\dot{x}_{\text{max}}|, \text{ where} \\
 A_d &= |Q_d[1]| \\
 B_d &= |R_d[1]| \\
 C_d &= \int_0^1 |R_d[u]| du
 \end{aligned}$$

where the absolute value applies element-wise to the vector b and to the vectors-of-polynomials Q_d, R_d .

A_d, B_d , and C_d are vectors of algebraic real numbers that depend only on the polynomial degree. Using the known closed form for the inverse of the Hilbert matrix, we can obtain expressions for P_d, Q_d , and R_d in the general case.

```

In[ ]:= Clear[d, Pd, Rd, Qd, Ad, Bd, Cd];
(*InverseHilbertMatrix[n_]:=Table[(-1)^(i+j) Binomial[n+i-1,n-j]
  Binomial[n+j-1,n-i] Binomial[i+j-2,i-1]^2,{i,1,n},{j,1,n}];
Pd[d_]:=InverseHilbertMatrix[d+1].Table[u^i,{i,0,d}]]//ExpandAll;*)
Pd[d_, i_] := Sum[(-1)^(i+j) (i+j+1) Binomial[d+i+1,d-j]
  Binomial[d+j+1,d-i] Binomial[i+j,i]^2 u^j, {j,0,d}];
Qd[d_, i_] := Integrate[Pd[d, i], {u, 0, u}];
Rd[d_, i_] := Integrate[Qd[d, i], {u, 0, u}];
Ad[d_] := Table[Abs[Qd[d, i]] /. {u -> 1}, {i, 0, d}];
Bd[d_] := Table[Abs[Rd[d, i]] /. {u -> 1}, {i, 0, d}];
Cd[d_] := Table[NIntegrate[Abs[Rd[d, i]], {u, 0, 1}], {i, 0, d}];

QdGeneralCase = Assuming[0 <= u && u <= 1, Unevaluated[Refine[Unevaluated[Qd[d, i]]]]]
RdGeneralCase = Assuming[0 <= u && u <= 1, Unevaluated[Refine[Unevaluated[Rd[d, i]]]]]

```

These expressions can then be specialized to any particular polynomial and coefficient orders d, j and integrated (numerically, in the case of C_d):

```

In[ ]:= dmax = 6;
Table[Ad[d], {d, 0, dmax}]
Table[Bd[d], {d, 0, dmax}]
Table[Cd[d], {d, 0, dmax}]
Table[
  Plot[Evaluate[Table[Rd[d, i], {i, 0, d}]], {u, 0, 1}, PlotRange -> Full], {d, 0, dmax}]

```

```

Out[ ]:= {{1}, {1, 0}, {1, 0, 0}, {1, 0, 0, 0},
  {1, 0, 0, 0, 0}, {1, 0, 0, 0, 0, 0}, {1, 0, 0, 0, 0, 0, 0}}

```

```


Out[ ]:= {{1/2}, {1, 1}, {1, 1, 0}, {1, 1, 0, 0},
  {1, 1, 0, 0, 0}, {1, 1, 0, 0, 0, 0}, {1, 1, 0, 0, 0, 0, 0}}

```

```

Out[ ]:= {{0.166667}, {0.416667, 0.5}, {0.5, 1., 0.5}, {0.5, 1., 1.16026, 0.729167},
  {0.5, 1., 2.08712, 3.18178, 1.562}, {0.5, 1., 3.32084, 8.87814, 9.94786, 3.94358},
  {0.5, 1., 4.89409, 19.8889, 37.4991, 32.9502, 10.925}}

```



```

In[ ]:= Assuming[d ∈ Integers && i ≥ 1 && i ∈ Integers,
  FullSimplify[(QdGeneralCase / (QdGeneralCase /. {i → 0})) /. {u → 1}]]

Out[ ]:= 
$$\frac{1}{-2 + 2 \Gamma[-d] \Gamma[2 + d]} (-1)^{1+i} (1 + i) \Gamma[-d]$$


$$\Gamma[2 + d] \left( (1 + i) \text{Binomial}[2 + d, d - i] \text{Binomial}[1 + d + i, -1 + d] \right.$$


$$\text{HypergeometricPFQ}[\{1 - d, 3 + d, 2 + i, 2 + i\}, \{3, 3 + i, 3 + i\}, 1] -$$


$$2 \text{Binomial}[1 + d, d - i] \text{Binomial}[1 + d + i, d]$$


$$\left. \text{HypergeometricPFQ}[-d, 2 + d, 1 + i, 1 + i], \{2, 2 + i, 2 + i\}, 1 \right)$$


```